Image J workshop 2019 material

Contributed by Yannick KREMPP
Monday, 17 June 2019
Last Updated Monday, 17 June 2019

As requested by the workshop attendees, here are the PDFs of the talk and the demo outline for those who want to reproduce some of the steps I have demoed onstage.

- Talk : link to the talk

- Demo outline: link to the outline

Here's also some of the macros used to process the demo sample (save them as a *.ijm file to use them in FIJI):

Converting a large image to a stack of smaller tiles :

```
macro "create stack of tiles [s]" {
//
run("RGB Color");
//ask for user input
n = getNumber("How many divisions?", 2);
id = getImageID();
title = getTitle();
getLocationAndSize(locX, locY, sizeW, sizeH);
```

```
width = getWidth();
height = getHeight();
tileWidth = width / n;
tileHeight = height / n;
newImage("TilesStack", "RGB composite-mode", tileWidth, tileHeight, 1, n*n, 1);
idStack = getImageID();
for (y = 0; y < n; y++) {
offsetY = y * height / n;
for (x = 0; x < n; x++) {
offsetX = x * width / n;
selectImage(id);
call("ij.gui.ImageWindow.setNextLocation", locX + offsetX, locY + offsetY);
tileTitle = "title="+ title + " [" + x + "," + y + "]";
makeRectangle(offsetX, offsetY, tileWidth, tileHeight);
run("Copy");
selectImage(idStack);
run("Paste");
run("Next Slice [>]");
}
}
selectImage(id);
close();
}
```

Processing a tile :

```
macro "Filter image and Analyze [f]" {
  // Make sure you don't touch the original image - useful for batch processing
  run("Duplicate...", "Copy of original image");
  run("RGB Color");
  // Ask for user input
  medianRadius = getNumber("Type in the median radius: ", 2);
  // Do the segmentation
  run("Median...", "radius=" + medianRadius);
  run("Color Threshold...");
  // Color Thresholder 2.0.0-rc-68/1.52h
  // Autogenerated macro, single images only!
  min = newArray(3);
  max = newArray(3);
  filter = newArray(3);
  a = getTitle();
  run("HSB Stack");
  run("Convert Stack to Images");
  selectWindow("Hue");
  rename("0");
  selectWindow("Saturation");
  rename("1");
  selectWindow("Brightness");
  rename("2");
  min[0] = 145;
  max[0] = 184;
  filter[0] = "pass";
  min[1] = 21;
  max[1] = 255;
  filter[1] = "pass";
  min[2] = 170;
  max[2] = 255;
  filter[2] = "pass";
  for (i = 0; i < 3; i++) {
    selectWindow("" + i);
    setThreshold(min[i], max[i]);
    run("Convert to Mask");
    if (filter[i] == "stop") run("Invert");
  }
  imageCalculator("AND create", "0", "1");
```

```
  imageCalculator("AND create", "Result of 0", "2");
  for (i = 0; i < 3; i++) {
    selectWindow("" + i);
    close();
  }
  selectWindow("Result of 0");
  close();
  selectWindow("Result of Result of 0");
  rename(a);
  // Colour Thresholding----------
  //setThreshold(255, 255);
  setOption("BlackBackground", true);
  run("Make Binary");
  // Clean the segmentation
  run("Median...", "radius=2");
  run("Fill Holes");
  run("Watershed");
  // Specify what has to be measured and analyze
  run("Set Measurements...", "area perimeter redirect=None decimal=3");
  run("Analyze Particles...", "size=50-Infinity display exclude clear add in_situ");
}
```

Yannick Krempp